Which conditions do the most useful drugs treat?

We found that the most useful drugs are Erlotinib, Bicalutamide, Leflunomide, Tamoxifen, and Methotrexate. This output shows the conditions that these drugs treat, in average usefulness order

```
call {

match (d:Drug)-[r:Assoc]->(c:Condition) with d,
sum(toInteger(r.usefulness))/sum(r.reviews) as
usefulness order by usefulness desc limit 5 return
collect(d) as drugs

}

match(c:Condition)<-[r]-(d:Drug) where d in drugs

return c.name as condition,
sum(toInteger(r.usefulness))/sum(r.reviews) as
usefulness

order by usefulness desc
```

	condition	usefulness
1	"Non-Small Cell Lung Cance"	80
2	"Rheumatoid Arthritis"	74
3	"Prostate Cance"	71
4	"Breast Cance"	58
5	"Breast Cancer, Prevention"	53

Which diseases have the largest number of different treatment options

```
match (d:Drug)-[r:Assoc]->(c:Condition)

return c.name as condition, count(distinct d.name) as drug_count

order by drug_count desc, condition
```

	condition	drug_count
1	"Breast Cancer, Metastatic"	6
2	"Chronic Myelogenous Leukemia"	6
3	"Non-Small Cell Lung Cance"	6
4	"Breast Cance"	4
5	"Renal Cell Carcinoma"	4
6	"Acute Lymphoblastic Leukemia"	3

```
call {

match (d:Drug)-[r:Assoc]->(c:Condition)

with c as condition, count(distinct d.name) as drug_count

order by drug_count desc limit 5

return collect(condition) as conditions

}

match (d:Drug)-[r:Assoc]->(c:Condition)

where c in conditions

return d,r,c
```



What drugs do the top 3 most-reviewed cancers have in common?

We found that the three most reviewed cancers are *Breast Cancer (Nonmetastatic)*, *Breast Cancer (Metastatic)*, and *Non-Small Cell Lung Cancer*. The two drugs that treat these cancers are Docetoxel and Pacitaxel, which both have the same target ('Microtubule stabiliser') and the same pathway ('Mitosis')




```
call {
match (d:Drug)-[r:Assoc]->(c:Condition)
where c.name contains 'Cance'
with c, sum(r.reviews) as nreviews order by nreviews desc limit 3
return collect(c) as top2c
}
match (oc:Condition)<-[r2]-(d:Drug)-[r1:Assoc]->(c:Condition)
where c in top2c and oc in top2c
return c,d,oc
```

	name	target	pathway
1	"PACLITAXEL"	"Microtubule stabiliser"	"Mitosis"
2	"PACLITAXEL"	"Microtubule stabiliser"	"Mitosis"
3	"DOCETAXEL"	"Microtubule stabiliser"	"Mitosis"

Which drugs have a large number of different uses?

We found that the 5 diseases with the largest number of different treatment options are Methotrexate, Cyclophosphamide, Imatinib, Sofarenib, and Temozolomide. Methotrexate, for example, treats 11 different conditions: Neoplastic Diseases, Rheumatoid Arthritis, Scleroderma, Gastric Cancer, Eczema, Cogan's Syndrome, Ectopic Pregnancy, Psoriatic Arthritis, Psoriasis, Lymphoma, and Systemic Sclerosis.

	drug	disease_count
1	"METHOTREXATE"	11
2	"CYCLOPHOSPHAMIDE"	5
3	"IMATINIB"	5
4	"SORAFENIB"	5
5	"TEMOZOLOMIDE"	5
6	"DASATINIB"	4

```
match (d:Drug)-[r:Assoc]->(c:Condition)

return d.name as drug, count(distinct c) as
disease_count

order by disease_count desc, drug
```

```
call {

match (d:Drug)-[r:Assoc]->(c:Condition)

with d, count(distinct c) as disease_count
order by disease_count desc limit 5

return collect(d) as drugs

}

match (d:Drug)-[r:Assoc]->(c:Condition)

where d in drugs

return d, r, c
```



What genes are most commonly targeted?

```
match (d:Drug)-[r:Assoc]->(c:Condition)
return d.target, count(d.target) as
count_drugs
order by count_drugs desc
```

	d.target	count_drugs
1	" KIT"	28
2	"ABL"	16
3	" PDGFR"	13
4	" VEGFR"	12
5	"PDGFR"	12
6	"Antimetabolite"	11

```
call {
match (d:Drug)-[r:Assoc]->(c:Condition)
with d.target as target, count(d.target) as drug_count
order by drug_count desc limit 1
return collect(target) as t
}
match (d:Drug)-[r:Assoc]->(c:Condition)
where d.target in t
return d, r, c
```



For the drugs that treat lung cancer, which other cancers do they treat?

```
match (c: Condition)<-[r:Assoc]-(d:Drug)-[r2:Assoc]->(c2: Condition)
where c2.name contains 'Lung'
return c.name, d.name
```



	c.name	d.name
1	"Ovarian Cance"	"PACLITAXEL"
2	"Breast Cancer, Metastatic"	"PACLITAXEL"
3	"Breast Cance"	"PACLITAXEL"
4	"Pancreatic Cance"	"ERLOTINIB"

What are the targets of the most useful drugs?

```
call {
match (d:Drug)-[r:Assoc]->(c:Condition)
with d ,sum(toInteger(r.usefulness))/sum(r.reviews) as usefulness
order by usefulness desc limit 5
return collect(d) as drugs
}
match(c:Condition)<-[r]-(d:Drug)
where d in drugs
return distinct d.target as target, count(d.target) as num_targets
order by num_targets desc
```

	target	num_targets
1	"Antimetabolite"	11
2	"ESR1"	4
3	"EGFR"	2
4	"Pyrimidine synthesis inhibitor"	2
5	"AR"	1

Which pathways do the most useful drugs treat by?

```
call {
match (d:Drug)-[r:Assoc]->(c:Condition)
with d ,sum(toInteger(r.usefulness))/sum(r.reviews) as usefulness
order by usefulness desc limit 5
return collect(d) as drugs
}
match(c:Condition)<-[r]-(d:Drug)
where d in drugs
return distinct d.pathway as pathway, count(d.pathway) as num_pathways
order by num_pathways desc
```

	pathway	num_pathways
1	"DNA replication"	13
2	"Hormone-related"	5
3	"EGFR signaling"	2

Which drugs have the average highest rated useful reviews?

```
match (d:Drug)-[r:Assoc]->(c:Condition)
return d.name as drug, sum(toInteger(r.usefulness))/sum(r.reviews) as avg_usefulness
order by avg_usefulness desc
```

	drug	avg_usefulness
1	"ERLOTINIB"	77
2	"BICALUTAMIDE"	71
3	"LEFLUNOMIDE"	63
4	"TAMOXIFEN"	56
5	"METHOTREXATE"	55